Using Units and Physical Constants in Mathematica

Mathematica has the ability to handle many calculations with units, and to perform conversions among different units. It also has many built-in constants and other data that are useful in doing calculations.

In[37]:=
```
Clear["Global`*"]
```

Basic Units

Imagine finding the volume of a box (2m) x (3m) x (5 cm). (Note the mixed units.) Enter the quantities in Mathematica using the ‘Quantity’ function. Mathematica will automatically suggest some units once you start typing. Note that units in Mathematica are always plural.

In[68]:=
```
l = Quantity[2., "Meters"]
```
Out[68]=
```
2. m
```

In[69]:=
```
w = Quantity[3., "Meters"]
```
Out[69]=
```
3. m
```

In[70]:=
```
h =  Quantity[5., "Centimeters"]
```
Out[70]=
```
5. cm
```

In[71]:=
```
volume = l * w * h
```
Out[71]=

$0.3\ \mathrm{m}^3$

You can convert to other volume units with UnitConvert.

In[72]:=
```
UnitConvert[volume, "Liters"]
```
Out[72]=
```
300. L
```

In[75]:=
```
UnitConvert[volume, "Centimeters"^3]
```
Out[75]=

$300\,000.\ \mathrm{cm}^3$

In[76]:=
```
UnitConvert[volume, "Gallons"]
```
Out[76]=
```
79.2516 gal
```

If you just want to convert to SI units, you can simply use “SI”:

In[77]:=
```
UnitConvert[volume, "SI"]
```
Out[77]=

$0.3\ \mathrm{m}^3$

Physical Constants

Mathematica has many constants built in, though sometimes you have to guess exactly how it will be expressed. For example, Boltzmann's constant k is:

In[45]:=
```
k = Quantity[1, "BoltzmannConstant"]
```
Out[45]=

k

Mathematica simply writes the standard letter *k* for this constant. To see what it is numerically, use UnitConvert[ ]:

In[46]:=
```
UnitConvert[k, "SI"]
```
Out[46]=

$$\frac{69\,032\,450}{801\,088\,317}\ \mathrm{meV/K}$$

There are two quirks with this answer. First, the set of units displayed is not always the most useful. If you know the units you want, you can request them directly. If you simply want the traditional SI base units (kg, m, s, A, K), use "SIBase".

In[47]:=
```
UnitConvert[k, "Joules" / "Kelvins"]
UnitConvert[k, "SIBase"]
```
Out[47]=

$$\frac{1\,380\,649}{100\,000\,000\,000\,000\,000\,000\,000\,000\,000}\ \mathrm{J/K}$$

Out[48]=

$$\frac{1\,380\,649}{100\,000\,000\,000\,000\,000\,000\,000\,000\,000}\ \mathrm{kg\,m^2/(s^2K)}$$

The second quirk is that for constants defined exactly in SI (such as the speed of light or Boltzmann's constant) Mathematica returns an exact integer result. You usually want to convert that to a numerical approximation with the N[ ] function.

In[49]:=
```
UnitConvert[N[k], "Joules" / "Kelvins"]
```
Out[49]=

$1.38065 \times 10^{-23}\ \mathrm{J/K}$

Physical and Chemical Data

Mathematica has access to a wealth of data. Some is built in, but most requires an on-line connection to fetch the data from Wolfram's servers.

For example, data about the elements is available through the ElementData function. Here is how to get the mass of a helium atom. The result is returned in terms of the atomic mass unit, but you can convert it to other forms:

In[50]:=
```
m = ElementData["Helium", "AtomicMass"]
```
Out[50]=

4.002602 u

In[51]:=
```
UnitConvert[m, "Kilograms"]
```
Out[51]=

6.646477×10^{-27} kg

In[52]:=
```
UnitConvert[m, "Megaelectronvolts" / "SpeedOfLightSquared"]
```
Out[52]=

3728.400 MeV/c^2

You can get a list of all available properties by using the “Properties” key:

In[53]:=
```
ElementData["Helium", "Properties"]
```

Data about chemical compounds is available through the ChemicalData[ ] function. For example, the mass of the water molecule is (where ‘u’ is the Atomic Mass Constant)

In[54]:=
```
mH2O = ChemicalData["Water", "MolecularMass"]
```
Out[54]=

18.015 u

In[83]:=
```
Quantity[1, "AtomicMassConstant"]
```
Out[83]=

m_u

In[84]:=
```
UnitConvert[%, "Kilograms"]
```
Out[84]=

$1.66053907 \times 10^{-27}$ kg

Calculations

Mathematica can do calculations with units and will perform appropriate operations on units. It will refuse to do calculations where the units do not agree.

Room Temperature

In[86]:=
```
k = Quantity[1, "BoltzmannConstant"]
```
Out[86]=

k

In[87]:=
```
T = Quantity[300., "Kelvins"]
```
Out[87]=
```
300. K
```

In[88]:=
```
UnitConvert[k T, "Joules"]
```
Out[88]=

4.14195×10^{-21} J

In[89]:=
```
UnitConvert[k T, "Electronvolts"]
```
Out[89]=
```
0.025852 eV
```

In[90]:=
```
k + T
```

Quantity : Kelvins and BoltzmannConstant are incompatible units

Quantity : Kelvins and BoltzmannConstant are incompatible units

Out[90]=
```
k + 300. K
```

For a longer example, consider problem 2.35. The final step is calculating a temperature from a complex combination of constants. In this problem, we are considering Helium initially at “room temperature” and “atmospheric pressure”.

In[60]:=

$$Tz = \frac{h^2}{2\, e^{5/3}\, k\, m\, \pi \left(\frac{k\, Ti}{pi}\right)^{2/3}}$$

Out[60]=

$$\frac{h^2}{2\, e^{5/3}\, k\, m\, \pi \left(\frac{k\, Ti}{pi}\right)^{2/3}}$$

In[91]:=
```
Ti = Quantity[300, "Kelvins"]
```
Out[91]=
```
300 K
```

In[92]:=
```
pi = Quantity[1, "Atmospheres"]
```
Out[92]=
```
1 atm
```

In[93]:=
```
k = Quantity[1, "BoltzmannConstant"]
```
Out[93]=
```
k
```

In[94]:=
```
h = Quantity[1, "PlanckConstant"]
```
Out[94]=
```
h
```

In[95]:=
```
m = ElementData["Helium", "AtomicMass"]
```
Out[95]=

$4.002602\,\mathrm{u}$

In[96]:=
```
Tz
```
Out[96]=

$0.0001675866\,\mathrm{atm}^{2/3}\mathrm{h}^2/(\mathrm{u}\,\mathrm{K}^{2/3}\mathrm{k}^{5/3})$

In[97]:=
```
UnitConvert[Tz, "Kelvins"]
```
Out[97]=

$0.01212015\,\mathrm{K}$